

Daten- und Kontrollfluss

Übersicht

- ◆ Datenfluss
 - ◆ Modusdeklaration
 - ◆ Unifikation als Filter und Konstruktor
- ◆ Kontrollfluss
 - ◆ Prozedurale vs. deklarative Semantik
 - ◆ Abstraktion, Sequenz, Alternation
- ◆ Elimination der Disjunktion
- ◆ Spezielle Lenkung des Kontrollflusses
 - ◆ Programmiertes Scheitern: fail/0
 - ◆ Nicht-Beweisbarkeit: \+/1
 - ◆ Suchbäume stützen: !/0
 - ◆ Aufruf: call/1

Modus: Instantiierung von Argumenten

Modusdeklarationen verdeutlichen, an welchen Argumentstellen Information hinein- bzw. herausfließt.

- ◆ Eingabe-Werte: Beim Aufruf instantiierte Variablen
 - ▶ Modusdeklaration: +
- ◆ Rückgabe-Werte: Nach Aufruf instantiierte Variablen
 - ▶ Modusdeklaration: -
- ◆ Wenn sowohl Ein- wie Rückgabe möglich ist:
 - ▶ Modusdeklaration: ?

```

% max(+Num,+Num,?Num)
max(X,Y,X):-
    X >= Y.
max(X,Y,Y):-
    X < Y.
    
```

```

?- max(2,4,Max).
Max = 4
    
```

```

?- max(2,4,2).
no
    
```

Unifikation: Filtern und Konstruieren

Doppelfunktion von Unifikation beim Beweisen

- ▶ **Filter:** Unifikation macht Fallunterscheidungen!
 - ◆ Nennt man auch *Pattern Matching* (Mustervergleich)
- ▶ **Konstruktor:** Unifikation macht automatischen Aufbau von Datenstrukturen!

Erst die Instantiierung beim Aufruf entscheidet, ob ein Argument Filter oder Konstruktor ist!

```

belegt(ida, vorlesung(ec11)).
belegt(ida, uebung(pc11)).
belegt(udo, vorlesung(pc11)).
belegt(udo, vorlesung(ec11)).

?- belegt(ida, X).
X = vorlesung(ec11) ;
X = uebung(pc11) ;
no

?- belegt(X, vorlesung(ec11)).
X = ida ;
X = udo ;
no
    
```

Prozedurale und deklarative Semantik



Deklarativ: Das Prädikat p ist ...

- ◆ wahr, falls q und r wahr ist.
- ◆ beweisbar, falls q und r beweisbar sind.

```

p :-
    q,
    r.
    
```

Prozedural: Um das Ziel/die Prozedur p ...


- ◆ zu erfüllen (*satisfy*), erfülle zuerst q und dann r.
- ◆ abzuarbeiten, rufe (*call*) zuerst q und dann r auf.

Prozedurale Interpretation

- ◆ hat relevante Reihenfolge!
- ◆ ist das, was Prolog-Interpreter kennt/berechnet.

Strukturierung des Kontrollflusses

Prädikatsdefinition, konjugierte Prädikate und mehrfache Klauseln stellen die elementaren, prozeduralen Kontrollen zur Verfügung:



Disjunktion

```
grossvater(Opa, Enkel) :-
  vater(Opa, Person),
  ( mutter(Person, Enkel)
  ; vater(Person, Enkel)
  ).
```

⇔

```
grossvater(Opa, Enkel) :-
  vater(Opa, Person),
  elter(Person, Enkel).
elter(Person, Kind) :-
  mutter(Person, Kind).
elter(Person, Kind) :-
  vater(Person, Kind).
```

Elimination der Disjunktion

- Disjunktionen können immer ersetzt werden, indem die disjunktiv verknüpften Terme zu Klauseln eines neuen Prädikats werden
- Disjunktionen sind wie Klauseldefinitionen: Es findet Backtracking statt inklusive Rückgängigmachen von Variablenbindungen!

Disjunktion und Backtracking

```
person(hans).
person(klara).
person(gabi).
person(kevin).
```

Programm

```
?- person(X).
X = hans ;
X = klara ;
X = gabi ;
X = kevin ;
no
```

Anfrage

Wie gehen wir vor, um alle Lösungen eines Ziels zu erhalten?

- bereits bekannt: Manuelle Eingabe eines Semikolons
- Anstatt der manuellen Nachfrage möchte man jedoch einen **programmierbaren Mechanismus** haben.

Programmiertes Backtracking: fail/0

Das eingebaute Prädikat fail/0 kann nie erfüllt werden!

- Backtracking kann damit programmiert werden
- Wichtig für Programmieretechnik *Failure-Driven-Loop*
 - Durch Scheitern lassen sich z.B. alle möglichen Lösungen ausgeben!

```
person(hans).
person(klara).
person(gabi).
person(kevin).
```

```
alle :-
  person(X),
  write(X), nl,
  fail.
```

Programm

```
?- alle.
no
```

Anfrage

```
hans
klara
gabi
kevin
```

Ausgabe

Failure-Driven Loop

Das Fehlschlagen der Anfrage: Ein behebarer Makel

- ◆ Eine zusätzliche, bedingungslos erfüllbare Klausel, die erst dann zum Zug kommt, wenn es keine weiteren Lösungen mehr gibt.

```

person(hans).
person(klara).
person(gabi).
person(kevin).
alle :-
  person(X),
  write(X), nl,
  fail.
alle.
    
```

Programm

```
?- alle.
yes
```

Anfrage

```

hans
klara
gabi
kevin
    
```

Ausgabe

Daten- und Kontrollfluss - 9

Nicht-Beweisbarkeit: \+ /1

Bisher konnten wir nur fragen, ob Prolog etwas beweisen kann:

```

person(hans).
person(klara).
    
```

- ◆ Ist Klara eine Person? `?- person(klara).`

Der Präfix-Operator \+ gelingt, falls sein Argument *nicht* bewiesen werden kann:

- ◆ Ist Fido keine Person? `?- \+ person(fido).`
yes
- ◆ Gibt es niemanden, der eine Person ist? `?- \+ person(Jemand).`
no

Daten- und Kontrollfluss - 10

Nicht-Beweisbarkeit und Negation

```

weiblich(X) :-
  \+ maennlich(X).
maennlich(hans).
    
```

```
?- weiblich(hans).
no
```

```
?- weiblich(gabi).
yes
```



Beachte: \+ ist keine logische Negation!

- ◆ Es gibt keine positive Information, dass Gabi weiblich ist.
- ◆ Gabi ist genauso weiblich wie Hermann nach diesem Programm.
- ◆ Je vollständiger ein Prädikat definiert ist, umso mehr nähert sich die Nicht-Beweisbarkeit der Negation an (*closed world assumption*).

```
?- weiblich(Wer).
no
```



```
?- weiblich(hermann).
yes
```

Daten- und Kontrollfluss - 11

Zwecklose Alternativen

```

max(X, Y, X) :-
  X >= Y.
max(X, Y, Y) :-
  X < Y.
    
```

```

?- max(2, 1, Max).
1 1 Call: max(2,1,_215) ?
2 2 Call: 2>=1 ?
2 2 Exit: 2>=1 ?
? 1 1 Exit: max(2,1,2) ?
X = 2 ? ;
1 1 Redo: max(2,1,2) ?
3 2 Call: 2<1 ?
3 2 Fail: 2<1 ?
1 1 Fail: max(2,1,_215) ?
no
    
```

Was geschieht bei der Anfrage?

- ◆ Beweisversuch mit erster Klausel, der auch gelingt
- ◆ Prolog merkt sich als Entscheidungspunkt die zweite Klausel. Es weiss nicht, dass immer nur eine einzige Klausel erfüllbar sein kann. D.H. es weiss nicht, dass die beiden Klauseln *deterministisch* sind!

Daten- und Kontrollfluss - 12

Cut !/0: Exklusive Fallunterscheidung

```
max2(X, Y, X) :-
  X >= Y,
  !.
max2(X, Y, Y) :-
  X < Y,
  !.
```

Mit dem eingebauten Prädikat !\0 lassen sich Klauseln als exklusive Fallunterscheidungen markieren.

► Der letzte Cut ist eigentlich unnötig!

Wirkung des Cut

- A. Wegschneiden aller alternativen Prädikats-Klauseln *unterhalb* jener, die den Cut enthält
- B. Wegschneiden aller alternativen Lösungen für Ziele, die in derselben Klausel *links* vom Cut stehen.

Wirkung des Cuts: Ausgangsprogramm

Abstraktes Beispielprogramm

♦ Die Anfrage hat 4 Lösungen.

```
p(1) :- q.
p(2).


q :- r(_).
q.

r(1).
r(2).
```

Programm

```
?- p(X).
X = 1 ? ;
X = 1 ? ;
X = 1 ? ;
X = 2 ? ;
no
```

Anfrage




Such-/Beweisbaum

Wirkung A des Cuts

cut/0 gelingt immer, aber als Nebeneffekt wird Suchbaum gestutzt.

- A. Wegschneiden aller alternativen Prädikats-Klauseln *unterhalb* jener, die den Cut enthält.



A im Such-/Beweisbaum

```
p(1) :- q.
p(2).

q :- r(_), !.
q.

r(1).
r(2).
```

A im Programm


Wirkung B des Cuts

cut/0 gelingt immer, aber als Nebeneffekt wird Suchbaum gestutzt.

- B. Wegschneiden aller alternativen Lösungen für Ziele, die in derselben Klausel *links* vom Cut stehen.

```
?- p(X).
X = 1 ? ;
X = 2 ? ;
no
```

Anfrage



B im Such-/Beweisbaum

```
p(1) :- q.
p(2).

q :- r(_), !!.
q.

r(1).
r(2).
```

B im Programm

Grüne vs. rote Cuts

Der Cut kann nur prozedural verstanden werden!

Grüne Cuts

- ◆ schneiden Suchäste ohne Lösungen weg.
- ◆ machen Programme effizienter (bei gleicher Lösungsmenge).
- ◆ zeigen oft Determinismus an (Kommentar `% green cut`).

Rote Cuts

- ◆ schneiden Suchäste mit unerwünschten Lösungen weg.
- ◆ machen Programme effizienter (bei veränderter Lösungsmenge).
- ◆ können Determinismus erzwingen.
- ◆ sind oft schlecht verständlich und heikel in der Verwendung (Kommentar `% red cut`)

Datenstrukturen zu Aufrufen: call/1

Das eingebaute Prädikat `call/1` gelingt, falls sein Argument bewiesen werden kann.

- ◆ Daten und Prozeduren sind keine strikt getrennten Welten.
- ◆ In der Anfrage `"?- call(person(hans))."` ist `call/1` redundant.
- ◆ `call/1` ist nur sinnvoll, wo das Argument variabel ist:

Zum Beispiel `once/1`:

Ein Prädikat, das sein Argument aufruft, aber höchstens eine Lösung erzeugt.

```
once(Goal) :-  
    call(Goal),  
    !.
```

Definition von `\+/1`

Unter Verwendung von `call/1`, `!/0` und `fail/0` lässt sich Nicht-Beweisbarkeit definieren.

```
:- op(900, fy, \+).  
\+ C :- call(C), !, fail.  
\+ C.
```

► Damit wird das Antwortverhalten bei `weiblich/1` erklärbar:

```
weiblich(X) :-  
    \+ maennlich(X).  
maennlich(hans).  
  
?- weiblich(Wer).  
no  
?- weiblich(hermann).  
yes
```